Polynômes multivariés et combinatoire

O. Marguin - 18 mai 2011

1. Fibonacci

```
> restart:
> F:=X/(1-X-X^2):
> series(F,X=0,21);
```

$$X+X^2+2X^3+3X^4+5X^5+8X^6+13X^7+21X^8+34X^9+55X^{10}+89X^{11}+144X^{12}+233X^{13}+377X^{14}+610X^{15}+987X^{16}+1597X^{17}+2584X^{18}+4181X^{19}+6765X^{20}+O(X^{21})$$

```
> fibo:=proc(n)
      local F,s;
      F:=X/(1-X-X^2);
      s:=series(F,X=0,n+1);
      coeff(s,X,n)
  end:
> fibo(20);
```

$$6765$$

```
> eq:=u(n)=u(n-1)+u(n-2);
```

$$eq := u(n) = u(n-1) + u(n-2)$$

```
> init:=u(0)=0,u(1)=1;
```

$$init := u(0) = 0,\ u(1) = 1$$

```
> s:=rsolve({eq,init},u(n));
```

$$s := \frac{\sqrt{5}\left(\frac{1}{2}+\frac{\sqrt{5}}{2}\right)^n}{5} - \frac{\sqrt{5}\left(\frac{1}{2}-\frac{\sqrt{5}}{2}\right)^n}{5}$$

```
> u20:=subs(n=20,s):factor(u20);
```

$$6765$$

2. Autre exemple de récurrence

```
> restart:
> eq:=a(n)=a(n-1)+2/n*a(n-2);
```

$$eq := a(n) = a(n-1) + \frac{2\,a(n-2)}{n}$$

```
> init:=a(0)=0,a(1)=1;
```

$$init := a(0) = 0,\ a(1) = 1$$

```
> rsolve({eq,init},a(n)): # donne un résultat peu sympathique
> aa[0]:=0:aa[1]:=1:s:=aa[0],aa[1]:
  for i from 2 to 20 do aa[i]:=aa[i-1]+2/i*aa[i-2];s:=s,aa[i] od:
  L:=[s];
```

$$L := \left[0, 1, 1, \frac{5}{3}, \frac{13}{6}, \frac{17}{6}, \frac{32}{9}, \frac{275}{63}, \frac{331}{63}, \frac{3529}{567}, \frac{20624}{2835}, \frac{262154}{31185}, \frac{899894}{93555}, \frac{13271546}{1216215}, \frac{104599444}{8513505}, \frac{159526664}{11609325}, \frac{3901834523}{255405150}, \frac{73350360107}{4341887550}, \frac{363242213927}{19538493975}, \frac{7561755305576}{371231385525}, \frac{82519155120373}{3712313855250}\right]$$

```
> with(gfun):
> guessgf(L,x);
```

$$\left[\frac{\left(-\frac{1}{4}(5-6x+2x^2)\,e^{(2x)}+\frac{5}{4}\right)e^{(-2x)}}{(-1+x)^3}, ogf\right]$$

```
> y:=op(1,%);
```

$$y := \frac{\left(-\frac{1}{4}(5-6x+2x^2)\,e^{(2x)}+\frac{5}{4}\right)e^{(-2x)}}{(-1+x)^3}$$

```
> series(y,x=0,20);
```

$$x+x^2+\frac{5}{3}x^3+\frac{13}{6}x^4+\frac{17}{6}x^5+\frac{32}{9}x^6+\frac{275}{63}x^7+\frac{331}{63}x^8+\frac{3529}{567}x^9+\frac{20624}{2835}x^{10}+\frac{262154}{31185}x^{11}+\frac{899894}{93555}x^{12}+\frac{13271546}{1216215}x^{13}+\frac{104599444}{8513505}x^{14}+\frac{159526664}{11609325}x^{15}+\frac{3901834523}{255405150}x^{16}+\frac{73350360107}{4341887550}x^{17}+\frac{363242213927}{19538493975}x^{18}+\frac{7561755305576}{371231385525}x^{19}+\mathrm{O}(x^{20})$$

```
> expand(y);
```

$$-\frac{\frac{5}{4}}{(-1+x)^3}+\frac{\frac{3}{2}x}{(-1+x)^3}-\frac{\frac{1}{2}x^2}{(-1+x)^3}+\frac{\frac{5}{4}}{(\mathbf{e}^x)^2(-1+x)^3}$$

On pourrait expliciter le terme général...

```
>
```

3. Problème du fût de bière

```
> restart:
```

On a f(1)=1, f(2)=2, f(3)=2.

f(n) est le nombre de triplets d'entiers naturels (d,s,f) tels que $d+2s+4f=n$. Donc f(400) vaut :

```
> t:=taylor(1/((1-x)*(1-x^2)*(1-x^4)),x=0,401): coeff(t,x,400);
```

$$10201$$

```
> L:=[seq(coeff(t,x,i),i=0..400)]:
> with(genfunc):
> formule:=rgf_findrecur(7,L,f,n);
```

$$formule := \mathrm{f}(n)=\mathrm{f}(n-1)+\mathrm{f}(n-2)-\mathrm{f}(n-3)+\mathrm{f}(n-4)-\mathrm{f}(n-5)-\mathrm{f}(n-6)+\mathrm{f}(n-7)$$

```
> with(linalg):
> A:=matrix(7,7,[0,1,0,0,0,0,0,0,0,1,0,0,0,0,0,0,0,1,0,0,0,0,0,0,0,1,0,0,0,0,0,0,0,1,
  0,0,0,0,0,0,0,1,1,-1,-1,1,-1,1,1]);
```

$$A := \begin{bmatrix} 0 & 1 & 0 & 0 & 0 & 0 & 0\\ 0 & 0 & 1 & 0 & 0 & 0 & 0\\ 0 & 0 & 0 & 1 & 0 & 0 & 0\\ 0 & 0 & 0 & 0 & 1 & 0 & 0\\ 0 & 0 & 0 & 0 & 0 & 1 & 0\\ 0 & 0 & 0 & 0 & 0 & 0 & 1\\ 1 & -1 & -1 & 1 & -1 & 1 & 1 \end{bmatrix}$$

```
> P:=minpoly(A,X);
```

$$P := -1+X+X^2-X^3+X^4-X^5-X^6+X^7$$

```
> r:=solve(P,X); # racines de P
```

$$r := -1, -1, I, -I, 1, 1, 1$$

On cherche le reste de la division euclidienne de X^n par P.

```
> eq:=X^n=P*Q(X)+ R;R:=add(a[i]*X^i,i=0..6);
```

$$eq := X^n=(-1+X+X^2-X^3+X^4-X^5-X^6+X^7)\,\mathrm{Q}(X)+a_0+a_1X+a_2X^2+a_3X^3+a_4X^4+a_5X^5+a_6X^6$$

$$R := a_0+a_1X+a_2X^2+a_3X^3+a_4X^4+a_5X^5+a_6X^6$$

```
> eqs:=subs(X=r[1],eq),subs(X=r[1],diff(eq,X)),subs(X=I,eq),subs(X=-I,eq),subs(X=1,eq
  ),subs(X=1,diff(eq,X)),subs(X=1,diff(eq,X,X));
```

$$eqs := (-1)^n=a_0-a_1+a_2-a_3+a_4-a_5+a_6,\ -(-1)^n n=a_1-2a_2+3a_3-4a_4+5a_5-6a_6,$$
$$I^n=a_0+a_1I-a_2-a_3I+a_4+a_5I-a_6,\ (-I)^n=a_0-a_1I-a_2+a_3I+a_4-a_5I-a_6,\ 1=a_0+a_1+a_2+a_3+a_4+a_5+a_6,$$
$$n=a_1+2a_2+3a_3+4a_4+5a_5+6a_6,\ n^2-n=2a_2+6a_3+12a_4+20a_5+30a_6$$

```
> vars:=coeffs(R,X);
```

$$vars := a_0, a_2, a_3, a_4, a_5, a_6, a_1$$

```
> s:=solve({eqs},{vars});
```

$$s := \{a_4=-\frac{13}{32}+\frac{(-1)^n n}{16}+\frac{(-1)^n}{32}+\frac{3I^n}{16}+\frac{3(-I)^n}{16}-\frac{n^2}{16}+\frac{7n}{16}-\frac{1}{16}I\,I^n+\frac{1}{16}I\,(-I)^n,$$
$$a_1=\frac{1}{2}-\frac{(-1)^n}{2}-\frac{1}{8}I\,I^n+\frac{1}{8}I\,(-I)^n-\frac{n}{8}+\frac{(-1)^n n}{8},\ a_3=\frac{1}{4}-\frac{(-1)^n}{4}+\frac{1}{4}I\,I^n-\frac{1}{4}I\,(-I)^n,$$
$$a_6=-\frac{(-I)^n}{16}-\frac{5(-1)^n}{32}+\frac{1}{16}I\,I^n-\frac{1}{16}I\,(-I)^n+\frac{9}{32}-\frac{5n}{16}+\frac{(-1)^n n}{16}+\frac{n^2}{16}-\frac{I^n}{16},$$

$$a_5 = \frac{1}{8} I\,(-I^n + (-I)^n + 2\,I - n\,I - 2\,I\,(-1)^n + (-1)^n\,n\,I),$$

$$a_0 = \frac{21}{32} - \frac{(-1)^n\,n}{16} + \frac{7\,(-1)^n}{32} + \frac{I^n}{16} + \frac{(-I)^n}{16} + \frac{n^2}{16} + \frac{1}{16} I\,I^n - \frac{1}{16} I\,(-I)^n - \frac{7\,n}{16},$$

$$a_2 = -\frac{n^2}{16} + \frac{5\,n}{16} - \frac{1}{32} + \frac{13\,(-1)^n}{32} - \frac{1}{16} I\,I^n + \frac{1}{16} I\,(-I)^n - \frac{(-1)^n\,n}{16} - \frac{3\,(-I)^n}{16} - \frac{3\,I^n}{16}\}$$

```
> assume(n,integer);
> Rsol:=evalc(Re(subs(s,R)));
```

$$Rsol := \frac{21}{32} - \frac{(-1)^{n\sim}\,n\sim}{16} + \frac{7\,(-1)^{n\sim}}{32} + \frac{n\sim^2}{16} - \frac{7\,n\sim}{16} + \frac{1}{8}\cos\left(\frac{n\sim\,\pi}{2}\right) - \frac{1}{8}\sin\left(\frac{n\sim\,\pi}{2}\right)$$
$$+ \frac{1}{16}\left(8 - 8\,(-1)^{n\sim} + 4\sin\left(\frac{n\sim\,\pi}{2}\right) - 2\,n\sim + 2\,(-1)^{n\sim}\,n\sim\right) X$$
$$+ \frac{1}{16}\left(-n\sim^2 + 5\,n\sim - \frac{1}{2} + \frac{13\,(-1)^{n\sim}}{2} + 2\sin\left(\frac{n\sim\,\pi}{2}\right) - (-1)^{n\sim}\,n\sim - 6\cos\left(\frac{n\sim\,\pi}{2}\right)\right) X^2 + \frac{1}{16}\left(4 - 4\,(-1)^{n\sim} - 8\sin\left(\frac{n\sim\,\pi}{2}\right)\right) X^3$$
$$+ \frac{1}{16}\left(-\frac{13}{2} + (-1)^{n\sim}\,n\sim + \frac{(-1)^{n\sim}}{2} + 6\cos\left(\frac{n\sim\,\pi}{2}\right) - n\sim^2 + 7\,n\sim + 2\sin\left(\frac{n\sim\,\pi}{2}\right)\right) X^4$$
$$+ \frac{1}{16}\left(4\sin\left(\frac{n\sim\,\pi}{2}\right) - 4 + 2\,n\sim + 4\,(-1)^{n\sim} - 2\,(-1)^{n\sim}\,n\sim\right) X^5$$
$$+ \frac{1}{16}\left(-2\cos\left(\frac{n\sim\,\pi}{2}\right) - \frac{5\,(-1)^{n\sim}}{2} - 2\sin\left(\frac{n\sim\,\pi}{2}\right) + \frac{9}{2} - 5\,n\sim + (-1)^{n\sim}\,n\sim + n\sim^2\right) X^6$$

```
> An:=evalm(subs(X=A,Rsol)):
> v0:=vector([1,1,2,2,4,4,6]):
> vn:=evalm(An &* v0):
```

D'où l'expression générale de $f(n)$:

```
> vn[1];
```

$$\frac{7\,n\sim}{16} + \frac{(-1)^{n\sim}\,n\sim}{16} + \frac{1}{8}\sin\left(\frac{n\sim\,\pi}{2}\right) + \frac{21}{32} + \frac{7\,(-1)^{n\sim}}{32} + \frac{n\sim^2}{16} + \frac{1}{8}\cos\left(\frac{n\sim\,\pi}{2}\right)$$

Vérifions :

```
> simplify(subs(n=400,vn[1]));
```

$$10201$$

Remarque : on aurait pu utiliser directement rsolve à partir de la formule de récurrence donnée par rgf_findrecur :

```
> n:='n':
> rsolve({formule,f(0)=1,f(1)=1,f(2)=2,f(3)=2,f(4)=4,f(5)=4,f(6)=6},f(n));
```

$$\frac{n}{4} + \frac{17}{32} + \frac{(n+1)\left(\frac{n}{2}+1\right)}{8} + \left(\frac{1}{16} + \frac{1}{16} I\right)(-I)^n + \left(\frac{1}{16} - \frac{1}{16} I\right) I^n + \left(\frac{n}{16} + \frac{1}{16}\right)(-1)^n + \frac{5\,(-1)^n}{32}$$

```
> assume(n,integer);sol:=simplify(%);
```

$$sol := \frac{7\,n\sim}{16} + \frac{21}{32} + \frac{n\sim^2}{16} + \frac{1}{16} e^{(-1/2\,I\,n\sim\,\pi)} + \frac{1}{16} I\,e^{(-1/2\,I\,n\sim\,\pi)} + \frac{1}{16} e^{(1/2\,I\,n\sim\,\pi)} - \frac{1}{16} I\,e^{(1/2\,I\,n\sim\,\pi)} + \frac{(-1)^{n\sim}\,n\sim}{16} + \frac{7\,(-1)^{n\sim}}{32}$$

```
> evalc(Re(convert(sol,trig)));
```

$$\frac{7\,n\sim}{16} + \frac{21}{32} + \frac{n\sim^2}{16} + \frac{1}{8}\cos\left(\frac{n\sim\,\pi}{2}\right) + \frac{1}{8}\sin\left(\frac{n\sim\,\pi}{2}\right) + \frac{(-1)^{n\sim}\,n\sim}{16} + \frac{7\,(-1)^{n\sim}}{32}$$

```
>
```

4. Nombres de Catalan

```
> restart:
> s:=solve(B=X+B^2,B);
```

$$s := \frac{1}{2} + \frac{\sqrt{1-4\,X}}{2}, \frac{1}{2} - \frac{\sqrt{1-4\,X}}{2}$$

```
> map(series,[s],X,11);
```

$$[1 - X - X^2 - 2\,X^3 - 5\,X^4 - 14\,X^5 - 42\,X^6 - 132\,X^7 - 429\,X^8 - 1430\,X^9 - 4862\,X^{10} + O(X^{11}),$$
$$X + X^2 + 2\,X^3 + 5\,X^4 + 14\,X^5 + 42\,X^6 + 132\,X^7 + 429\,X^8 + 1430\,X^9 + 4862\,X^{10} + O(X^{11})]$$

C'est la seconde solution qui convient.

```
> seq(1/n*binomial(2*n-2,n-1),n=1..10);
```

$$1, 1, 2, 5, 14, 42, 132, 429, 1430, 4862$$

5. Décompositions de Lagrange

```
> restart:
> Lagrange:=proc(n)
      local C,k;
      C:=add(X^(k^2),k=0..isqrt(n));
      coeff(C^4,X,n);
  end:
> Lagrange(1000);
```

$$276$$

6. Problème des reines

```
> restart:
> n:=4;
```

$$n := 4$$

```
> P:=product(product((1+l[i]*c[j]*d[i+j-1]*e[i-j+n]),j=1..n),i=1..n); # formule (5) du cours
```

$$P := (1+l_1 c_1 d_1 e_4)(1+l_1 c_2 d_2 e_3)(1+l_1 c_3 d_3 e_2)(1+l_1 c_4 d_4 e_1)(1+l_2 c_1 d_2 e_5)(1+l_2 c_2 d_3 e_4)(1+l_2 c_3 d_4 e_3)(1+l_2 c_4 d_5 e_2)(1+l_3 c_1 d_3 e_6)(1+l_3 c_2 d_4 e_5)(1+l_3 c_3 d_5 e_4)(1+l_3 c_4 d_6 e_3)(1+l_4 c_1 d_4 e_7)(1+l_4 c_2 d_5 e_6)(1+l_4 c_3 d_6 e_5)(1+l_4 c_4 d_7 e_4)$$

```
> P:=factor(coeff(series(P,l[1],2),l[1],1));
```

$$P := (c_4 d_4 e_1 + c_3 d_3 e_2 + c_2 d_2 e_3 + c_1 d_1 e_4)(1+l_2 c_1 d_2 e_5)(1+l_2 c_2 d_3 e_4)(1+l_2 c_3 d_4 e_3)(1+l_2 c_4 d_5 e_2)(1+l_3 c_1 d_3 e_6)(1+l_3 c_2 d_4 e_5)(1+l_3 c_3 d_5 e_4)(1+l_3 c_4 d_6 e_3)(1+l_4 c_1 d_4 e_7)(1+l_4 c_2 d_5 e_6)(1+l_4 c_3 d_6 e_5)(1+l_4 c_4 d_7 e_4)$$

```
> P:=factor(coeff(series(P,l[2],2),l[2],1));
```

$$P := (c_3 d_4 e_3 + c_2 d_3 e_4 + c_1 d_2 e_5 + c_4 d_5 e_2)(c_4 d_4 e_1 + c_3 d_3 e_2 + c_2 d_2 e_3 + c_1 d_1 e_4)(1+l_3 c_1 d_3 e_6)(1+l_3 c_2 d_4 e_5)(1+l_3 c_3 d_5 e_4)(1+l_3 c_4 d_6 e_3)(1+l_4 c_1 d_4 e_7)(1+l_4 c_2 d_5 e_6)(1+l_4 c_3 d_6 e_5)(1+l_4 c_4 d_7 e_4)$$

```
> P:=factor(coeff(series(P,l[3],2),l[3],1));
```

$$P := (c_3 d_5 e_4 + c_2 d_4 e_5 + c_1 d_3 e_6 + c_4 d_6 e_3)(c_3 d_4 e_3 + c_2 d_3 e_4 + c_1 d_2 e_5 + c_4 d_5 e_2)(c_4 d_4 e_1 + c_3 d_3 e_2 + c_2 d_2 e_3 + c_1 d_1 e_4)(1+l_4 c_1 d_4 e_7)(1+l_4 c_2 d_5 e_6)(1+l_4 c_3 d_6 e_5)(1+l_4 c_4 d_7 e_4)$$

```
> P:=factor(coeff(series(P,l[4],2),l[4],1));
```

$$P := (c_4 d_7 e_4 + c_3 d_6 e_5 + c_2 d_5 e_6 + c_1 d_4 e_7)(c_3 d_5 e_4 + c_2 d_4 e_5 + c_1 d_3 e_6 + c_4 d_6 e_3)(c_3 d_4 e_3 + c_2 d_3 e_4 + c_1 d_2 e_5 + c_4 d_5 e_2)(c_4 d_4 e_1 + c_3 d_3 e_2 + c_2 d_2 e_3 + c_1 d_1 e_4)$$

```
> PP:=mul(add(c[j]*d[i+j-1]*e[i-j+n],j=1..n),i=1..n); # formule (6) du cours
```

$$PP := (c_4 d_7 e_4 + c_3 d_6 e_5 + c_2 d_5 e_6 + c_1 d_4 e_7)(c_3 d_5 e_4 + c_2 d_4 e_5 + c_1 d_3 e_6 + c_4 d_6 e_3)(c_3 d_4 e_3 + c_2 d_3 e_4 + c_1 d_2 e_5 + c_4 d_5 e_2)(c_4 d_4 e_1 + c_3 d_3 e_2 + c_2 d_2 e_3 + c_1 d_1 e_4)$$

```
> simplify(P-PP); # vérification
```

$$0$$

```
> P:=factor(coeff(series(P,c[1],2),c[1],1)):
> P:=factor(coeff(series(P,c[2],2),c[2],1)):
> P:=factor(coeff(series(P,c[3],2),c[3],1)):
> P:=factor(coeff(series(P,c[4],2),c[4],1)):
> P:=remove(hastype,P,`^`);
```

$$P := 2\, d_2 e_5 d_5 e_6 d_6 e_3 d_3 e_2$$

Donc la réponse est 2. Automatisons ce calcul pour n quelconque.

```
> reines:=proc(n)
      local i,j,P,c,d,e,L;
      P:=mul(add(c[j]*d[i+j-1]*e[i-j+n],j=1..n),i=1..n);
      for i to n do P:=coeff(series(P,c[i],2),c[i],1) od;
      P:=expand(P);
      P:=remove(hastype,P,`^`);
      subs({seq(e[i]=1,i=1..2*n-1),seq(d[j]=1,j=1..2*n-1)},P)
  end:
> seq(reines(n),n=4..8);
```

2, 10, 4, 40, 92

Chronométrons pour $n = 9$:

```
> debut:=time():reines(9);time()-debut;
```

352

9.516

On ne peut raisonnablement pas aller au delà de n=9 par cette méthode.

Autre stratégie classique : backtracking

```
> restart:
> reinesback:=proc(n)
      local essai,test,cpt,x;
      test:=proc(k)
         local i;
         for i from 1 to k-1 do
            if x[i]=x[k] or abs(x[i]-x[k])=k-i then
               return false
            fi
         od;
         return true
      end;
      essai:=proc(k)
         for x[k] from 1 to n do
            if test(k) then
               if k=n then
                  cpt:=cpt+1
               else
                  essai(k+1)
               fi
            fi
         od
      end;
      cpt:=0;
      essai(1);
      cpt
  end:
> seq(reinesback(n),n=4..8);
```

2, 10, 4, 40, 92

Chronométrons pour $n = 9$:

```
> debut:=time():reinesback(9);time()-debut;
```

352

1.219

```
> reinesback(10);
```

724

7. Somme des entiers balancés de longueur 2n

```
> restart:
> E:=proc(n)  # calcule En
      local V,W,S,G,s,g;
      V:=add(Y^r,r=0..9);
      W:=add(r*Y^r,r=0..9);
      S:=X*W/((1-X*V)*(1-10*X*V));
      G:=V*X/(1-V*X);
      s:=series(S,X,n+1);
      g:=series(G,X,n+1);
      (10^n+1)*add(coeff(coeff(s,X,n),Y,k)*coeff(coeff(g,X,n),Y,k),k=0..9*n);
  end:
> seq(E(n),n=1..20);
```

495, 3349665, 27625972374, 240801497591985, 2162288199783771180, 19790585209980209414790,
183566563673998164334363260, 1719500099286549828049990071345, 16229128291876975983770871708123024,
154095946187094841998459040538129051580, 1470283473289020339999852971652671097966000,
14085156389112875121049985914843610887124878950,

135392958469359073392604998646070415306409266073950,
1305233650788904309294105679869476634921109569070589432,
12614395930501727321029630695987385604069498272678970369304,
122177670644611024459763947361498778223293553889755402360526385,
1185633898500558643116053969483499881436610149944135688394603051650,
11525195623906119101843912373578899988474804376093880898156087626421100,
112203312767859412537217211281711779998877966872321405874627827887182882200,
1093844474149520613133628019194480743399890615552585047938686637198080551925660
>